# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=servant-client
pkgname=haskell-servant-client
pkgver=0.19
pkgrel=83
pkgdesc="Automatic derivation of querying functions for servant"
url="http://docs.servant.dev/"
license=("BSD")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-base-compat' 'haskell-http-client' 'haskell-http-media'
         'haskell-http-types' 'haskell-kan-extensions' 'haskell-monad-control'
         'haskell-semigroupoids' 'haskell-servant' 'haskell-servant-client-core'
         'haskell-transformers-base' 'haskell-transformers-compat')
makedepends=('ghc' 'uusi' 'haskell-hunit' 'haskell-quickcheck' 'haskell-aeson' 'haskell-entropy'
             'haskell-hspec' 'haskell-hspec-discover' 'haskell-http-api-data'
             'haskell-markdown-unlit' 'haskell-network' 'haskell-servant-server' 'haskell-sop-core'
             'haskell-tdigest' 'haskell-wai' 'haskell-warp')
source=("https://hackage.haskell.org/packages/archive/$_hkgname/$pkgver/$_hkgname-$pkgver.tar.gz")
sha512sums=('a8293795ccea98a83454b663d670cf98c07aea67afc69a1c618bacb3de0ddda672c7abc11576f9681674a4c1fe8c0375b294beeeb6e47a269621f5f11f492925')

prepare() {
  cd $_hkgname-$pkgver
  uusi -u base-compat -u hspec $_hkgname.cabal
}

build() {
  cd $_hkgname-$pkgver

  runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
    --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
    --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
    --ghc-option=-optl-Wl\,-z\,relro\,-z\,now \
    --ghc-option='-pie'

  runhaskell Setup build $MAKEFLAGS
  runhaskell Setup register --gen-script
  runhaskell Setup unregister --gen-script
  sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
  sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
  cd $_hkgname-$pkgver
  runhaskell Setup test --show-details=direct
}

package() {
  cd $_hkgname-$pkgver

  install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
  install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
  runhaskell Setup copy --destdir="$pkgdir"
  install -D -m644 LICENSE -t "$pkgdir"/usr/share/licenses/$pkgname/
  rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
